# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=smbspider
pkgver=10.7db9323
pkgrel=5
groups=('blackarch' 'blackarch-scanner')
pkgdesc='A lightweight python utility for searching SMB/CIFS/Samba file shares.'
arch=('any')
url='https://github.com/T-S-A/smbspider'
license=('GPL2')
depends=('python2' 'python2-netaddr' 'python2-pysmbc' 'python2-pysmb')
makedepends=('git')
source=("git+https://github.com/T-S-A/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

prepare() {
  cd $pkgname

  sed -i 's|/usr/bin/python|/usr/bin/python2|' "$pkgname.py"
}

package() {
  cd $pkgname

  install -Dm 755 "$pkgname.py" "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md
  install -Dm 644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

